Premiers pas en Maple *(Partie 5 de 5)*

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Pour voir le contenu des différentes sections, cliquer avec la souris sur le triangle ▶ précédant le titre. La section se déploiera ▼ et son contenu sera affiché.
(un clic gauche ou un clic droit sur le triangle permettra de rétracter la section et son contenu sera alors masqué).

Pour votre confort, vous pouvez ajuster la taille de l'affichage à l'aide de la commande *Facteur de zoom* du menu *Affichage*.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,display,spacecurve);
```

$$[display, spacecurve] \tag{1.1}$$

Objectif

Cette cinquième partie a comme objectif de présenter les macro-commandes pour

- un calcul d'une limite d'une fonction d'une et de deux variables;
- un calcul de la dérivée explicite et implicite d'une fonction d'une seule variable
  - en utilisant l'opérateur de dérivation `D`
  - utilisant les macro-commandes `diff` et `implicitdiff`;
- un calcul des dérivées partielles
  - d'une fonction explicite de deux variables
  - d'une fonction implicite de deux variables
- une intégration simple et une intégration double

Le calcul de limites d'une fonction d'une seule variable

La macro-commande limit est à utiliser pour faire le calcul des limites. Les options **left** et **right** permettent d'effectuer un calcul de limite directionnelle.

```
> limit((x^3-4*x^2+3*x)/(x-3),x=3,left);
```

$$6 \tag{3.1}$$

On documente beaucoup mieux un développement de calcul en uilisant la forme inerte de la macro-commande limit.

```
> f:=x->(x^3-4*x^2+3*x)/(x-3);
```

$$f := x \mapsto \frac{x^3 - 4 \cdot x^2 + 3 \cdot x}{x - 3} \qquad \textbf{(3.2)}$$

```
> Limit(f(x),x=3,left)=limit(f(x),x=3,left);
```

$$\lim_{x \to 3-} \frac{x^3 - 4x^2 + 3x}{x - 3} = 6 \qquad \textbf{(3.3)}$$

```
> f:=x->x*sin(1/x);
```

$$f := x \mapsto x \cdot \sin\left(\frac{1}{x}\right) \qquad \textbf{(3.4)}$$

```
> Limit(f(x),x=0)=limit(f(x),x=0);
```

$$\lim_{x \to 0} x \sin\left(\frac{1}{x}\right) = 0 \qquad \textbf{(3.5)}$$

Maple effectue automatiquement les simplifications sur le corps des nombres complexes. Il faut donc bien faire attention à poser seulement des calculs pertinents de limites: il faut s'assurer que la fonction est toujours définie et réelle dans tout voisinage de la limite demandée.

```
> f:=x->(sqrt(1-x)+x)/(1-x);
```

$$f := x \mapsto \frac{\sqrt{1 - x} + x}{1 - x} \qquad \textbf{(3.6)}$$

```
> Limit(f(x),x=1,right);
```

$$\lim_{x \to 1+} \frac{\sqrt{1 - x} + x}{1 - x} \qquad \textbf{(3.7)}$$

```
> value((3.7));
```

$$-\infty \qquad \textbf{(3.8)}$$

Comment faut-il interpréter cette réponse ? La fonction f n'est pas définie dans $\mathbb{R}$ lorsque $x > 1$.

Un calcul d'une limite infinie.

```
> f:=x->(3*x^2+1)/(5*x^2-7);
```

$$f := x \mapsto \frac{3 \cdot x^2 + 1}{5 \cdot x^2 - 7} \qquad \textbf{(3.9)}$$

```
> Limit(f(x),x=infinity)=limit(f(x),x=infinity);
```

$$\lim_{x \to \infty} \frac{3x^2 + 1}{5x^2 - 7} = \frac{3}{5} \qquad \textbf{(3.10)}$$

Le calcul de limites d'une fonction de deux variables

Cas où la limite existe

Soit la fonction f définie par $f(x, y) = 10 + x y^2$.

On a que $dom_f = \mathbb{R}\times\mathbb{R}$.

Traçons cette fonction sur un pavé circulaire centré à l'origine de rayon 5, soit pour $x \in [-5, 5]$ et pour $y \in \left[-\sqrt{-x^2+25}, \sqrt{-x^2+25}\right]$.

```
> f:=(x,y)->10+x*y^2;
```

$$f := (x, y) \mapsto 10 + x \cdot y^2 \qquad \textbf{(4.1.1)}$$

```
> Surface:=plot3d([x,y,f(x,y)],x=-5..5,y=-sqrt(25-x^2)..sqrt(25-x^2),
              grid=[60,30],
              axes=boxed,linestyle=0):
> Vue:=display(Surface):
> display(Vue,shading=ZHUE,axes=framed);
```



Sélectionnez le graphique et en tenant le bouton gauche de la souris pressé, modifiez l'orientation du graphique. La perception de la troisième dimension sera à son maximum. L'orientation [0,0,0] est une vue en plongée vertcicale montrant que le domaine du tarcé est bien circulaire.

Illustrons graphiquement l'existence de la limite de cette fonction lorsque $(x, y) \rightarrow (1, 2)$.

Traçons à nouveau cette fonction mais sur un pavé rectangulaire pour illustrer notre propos. Soit donc pour $x \in [-2, 3]$ et pour $y \in [-2, 3]$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-2..3,y=-2..3,
              grid=[20,20],
```

```
                axes=normal,linestyle=2):
  Vue_1:=display(Surface):
> display(Vue_1,shading=ZHUE,orientation=[30,60]);
```



Puisque $dom_f = \mathbb{R}\times\mathbb{R}$, quelque soit le vosinage de (1,2), la fonction f est toujours définie.
Superposons au graphique précédent, un voisinage circulaire centré en (1,2) et réalisons le tracé de l'image de ce voisinage:

- ce voisinage sera visualisé par un cercle centré au point (1,2,0) dans le plan XY
- l'image de ce voisinage sera visualisé par la portion intérieure de la surface limitée par l'image du périmètre du voisinage de (1,2,0).

```
> Voisinage:=spacecurve([1+cos(t)/4,2+sin(t)/4,0],t=0..2*Pi,color=
  orange,thickness=2):
  Image:=spacecurve([1+cos(t)/4,2+sin(t)/4,f(1+cos(t)/4,2+sin(t)/4)
  ],t=0..2*Pi,
          color=orange,
          thickness=2):
  Horizontale:=spacecurve([1,t,0],t=0..2,color=black,thickness=1):
  Verticale:=spacecurve([t,2,0],t=0..1,color=black,thickness=1):
> Vue_2:=display([Vue_1,Voisinage,Image,Horizontale,Verticale]):
> display(Vue_2,shading=ZHUE,orientation=[25,77,0]);
```



Illustrons maintenant la projection de ces images sur le plan XZ et bornons, dans le plan XZ, la cote de ces images entre les traces des plans d'équations $z = 14 - \epsilon$ et $z = 14 + \epsilon$ en prenant $\epsilon = 2$.

```
> Proj_xz:=spacecurve([1+cos(t)/4,0,f(1+cos(t)/4,2+sin(t)/4)],t=0.
  .2*Pi,
          color=orange,
          thickness=2):
  Verticale:=spacecurve([1,2,t],t=0..f(1,2),linestyle=2,thickness=1,
  color=orange):
  Horizontale:=spacecurve([1,t,f(1,2)],t=0..2,linestyle=2,thickness=
  1,color=orange):
  L:=spacecurve([t,0,14],t=0..3,thickness=2,color=orange,linestyle=
  2):
  epsilon:=2:
  L_moins_delta:=spacecurve([t,0,14-epsilon],t=0..3,thickness=2,
  color=khaki,linestyle=2):
  L_plus_delta:=spacecurve([t,0,14+epsilon],t=0..3,thickness=2,
  color=khaki,linestyle=2):
> Vue_3:=display([Vue_2,L,L_moins_delta,L_plus_delta,Proj_xz,
  Verticale,Horizontale]):
> display(Vue_3,shading=ZHUE,orientation=[45,60,0]);
```



Effectuons un zoom vers l'avant pour « mieux voir »

```
> display(Vue_3,shading=ZHUE,view=[0..3,0..3,0..20],orientation=[45,
  60,0]);
```



Visualisons la courbe sur la surface entre les plans d'équations $z = 14 - \epsilon$ et $z = 14 + \epsilon$

```
> Plan_sup:=plot3d([x,y,14+epsilon],x=0..3,y=0..3,style=WIREFRAME,
  color=wheat):
  Plan_inf:=plot3d([x,y,14-epsilon],x=0..3,y=0..3,style=PATCHNOGRID,
  color=wheat):
  Vue_4:=display([Voisinage,Image,L,L_moins_delta,L_plus_delta,
  Proj_xz,Verticale,Horizontale],
  Plan_sup,Plan_inf):
> display(Vue_4,shading=ZHUE,Surface,axes=normal,orientation=[27,87,
  0]);
```



Effectuons un zoom vers l'avant et *n'affichons pas la surface* afin de mieux visualiser les valeurs d'images qui sont comprises entre ces deux plans.

```
> display(Vue_4,axes=normal,orientation=[26,82]);
```



On voit mieux maintenant que pour tout $\epsilon > 0$, il sera toujours possible d'obtenir un voisinage circulaire de (1,2,0) de telle manière que les cotes de toutes les images de ce voisinage seront comprises entre les plans d'équations $z = 14 - \epsilon$ et d'équation $z = 14 + \epsilon$.

Évaluons finalement cette limite avec Maple.

```
> Limit(f(x,y),{x=1,y=2})=limit(f(x,y),{x=1,y=2});
```

$$\mathit{Limit}\left(x y^2 + 10, \{x = 1, y = 2\}\right) = 14 \quad \textbf{(4.1.2)}$$

REMARQUE: Dans le cas d'une fonction où l'image d'un voisinage troué $W_0(a, b)$ ne serait pas plane (le cas où il y aurait des «bosses») , il est clair que la projection dans le plan XZ de l'image de la "frontière de $W_0(a, b)$" pourrait ne pas contenir le maximum et/ou le minimun de l'image. Dans de tels cas, cette projection dans le plan XZ ne serait pas appropriée pour ce voisinage $W_0(a, b)$. Mais, il n'en demeure pas moins, lorsque la limite existe, qu'il sera toujours possible d'obtenir un $W_0(a, b)$ de telle manière que l'image de ce voisinage soit tout entier compris entre les plans d'équations $z = 14 - \epsilon$ et $z = 14 + \epsilon$, et ce, quelque soit $\epsilon > 0$.

Cas où la limite n'existe pas

Cas où la limite n'existe pas par défaut

Soit la fonction f définie par $f(x) = \dfrac{x^2 + y^2}{x + y}$.

Illustrons graphiquement que la limite de cette fonction lorsque $(x, y) \to (0, 0)$ n'existe pas par défaut. En effet, puisque le $dom_f = \mathbb{R} \times \mathbb{R} \setminus \{(x,y) \mid y = -x\}$, tout voisinage de (0,0) possède des points de la droite d'équation $y = -x$. Et donc, que la limite n'existe pas.

Illustrons graphiquement qu'il n'existe aucun voisinage de (0,0) pour lequel la fonction est toujours définie.

```
> f:=(x,y)->(x^2+y^2)/(x+y);
```

$$f := (x, y) \mapsto \frac{y^2 + x^2}{x + y} \qquad \textbf{(4.2.1.1)}$$

Traçons la surface correspondant à l'équation $f(x) = \dfrac{x^2 + y^2}{x + y}$ pour $x \in [-0,25;\ 0,25]$ et pour $y \in [-0,25;\ 0,25]$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.25..0.25,y=-0.25..0.25,
           axes=normal,linestyle=2,
           grid=[20,20]):
  display(Surface,shading=zhue,axes=normal,orientation=[-25,40]);
```



On constate donc une « faille » dans la direction du plan d'équation $y = -x$. (Sélectionnez le graphique et obtenez ensuite différentes orientations de cette surface avec l'outil pivotement).

Après avoir visualisé différentes orientations de cette surface et compte tenu de la manière dont l'afficheur représente cette surface, il semble qu'il y a des voisinages appropriés de (0,0) où la fonction f est définie. Cela est dû au fait que lorsque y est à peu près égale à $-x$, la fonction f prend de très petites valeurs (de l'ordre de $(-10)^{15}$) et que l'afficheur doit composer en même temps avec des valeurs d'abscisses et d'ordonnées près de zéro.

En fait, cette « faille » n'est pas graphiquement réaliste.

Améliorons le tracé de cette surface par un contrôle de l'affichage avec l'option `view`.

```
> display(Surface,shading=zhue,orientation=[-40,75,0],
         axes=normal,view=[-0.25..0.25,-0.25..0.25,-2..2]);
```



Améliorons encore plus ce tracé avec un `grid` supérieur (un plus grand nombre de points est calculé dans ce cas).

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.25..0.25,y=-0.25..0.25,
          axes=normal,linestyle=2,
          grid=[60,60]):
> display(Surface,shading=zhue,orientation=[-40,75,0],
         axes=normal,view=[-0.25..0.25,-0.25..0.25,-5..5]);
```



En augmentant la "résolution", le tracé est davantage près du tracé correct de cette surface. Dans le plan $y = -x$, il n'y a toujours pas de points.

Mais ce contrôle de l'axe des z est-il judicieusement fait pour qu'on puisse " voir " correctement cette surface ? La surface semble avoir une "courbure" qui a été tronquée par un contrôle insuffisant de la cote. Peaufinons davantage le tracé en diminuant l'intervalle de valeurs de la cote dans l'option `view`.

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.25..0.24,y=-0.25..0.24,
           linestyle=1,
           grid=[60,60]):
> display(Surface,orientation=[-15,66,0],
          shading=zhue,axes=boxed,
          scaling=constrained,
          view=[-0.25..0.25,-0.25..0.25,-0.18..0.18]);
```



(Pour mieux voir la surface obtenue, sélectionnez le graphique et agrandissez avec l'outil loupe +)

Dans ce dernier tracé, la « faille » s'est transformée en un pseudo-plan d'équation $y = -x$. Malheureusement, pour les tracé en 3D, on ne dispose pas de l'option `discont=true` qui éliminerait les pseudo-plans comme pour les tracés en 2D où cela élimine les pseudo-asymptotes. Avec une telle option, le tracé aurait été fait avec plus de discernement en évitant de relier les points dont les cotes extrêmes sont obtenues avec des valeurs de y presqu'égales à celles de $-x$.

En fait, le tracé correct de cette surface est une surface dont toutes sections parallèles au plan XY est un cercle quasi-tangent au plan d'équation $y = -x$. En effet, résolvons l'équation $f(x, y) = k$:

```
> Éq_1:=f(x,y)=k;
```

$$\acute{E}q\_1 := \frac{x^2 + y^2}{x + y} = k \qquad \textbf{(4.2.1.2)}$$

```
> Éq_2:=(x+y)*Éq_1;
```

$$\acute{E}q\_2 := x^2 + y^2 = (x + y)\, k \qquad \textbf{(4.2.1.3)}$$

```
> Éq_3:=expand(Éq_2-(rhs(Éq_2)=rhs(Éq_2)));
```

$$\acute{E}q\_3 := -kx - ky + x^2 + y^2 = 0 \qquad \textbf{(4.2.1.4)}$$

```
> Éq_4:=Student[Precalculus][CompleteSquare](Éq_3,x);
```

$$\acute{E}q\_4 := \left(x - \frac{k}{2}\right)^2 - ky + y^2 - \frac{k^2}{4} = 0 \qquad \textbf{(4.2.1.5)}$$

```
> Éq_5:=Student[Precalculus][CompleteSquare](Éq_4,y);
```

$$Éq\_5 := \left(y-\frac{k}{2}\right)^2+\left(x-\frac{k}{2}\right)^2-\frac{k^2}{2}=0 \quad \textbf{(4.2.1.6)}$$

```
> Éq_6:=Éq_5+(1/2*k^2=1/2*k^2);
```

$$Éq\_6 := \left(y-\frac{k}{2}\right)^2+\left(x-\frac{k}{2}\right)^2=\frac{k^2}{2} \quad \textbf{(4.2.1.7)}$$

On a donc, pour toute " hauteur" $k \in \mathbb{R} \setminus \{0\}$, nous avons un cercle centré au point $\left(\frac{k}{2}, \frac{k}{2}\right)$ de rayon $\frac{\sqrt{2}\,|k|}{2}$ excluant tous $(x,-x)$.

```
> Éq_7:=(subs(y=-x,Éq_6));
```

$$Éq\_7 := \left(-x-\frac{k}{2}\right)^2+\left(x-\frac{k}{2}\right)^2=\frac{k^2}{2} \quad \textbf{(4.2.1.8)}$$

```
> solve(Éq_7,{x});
```

$$\{x=0\}, \{x=0\} \quad \textbf{(4.2.1.9)}$$

L'unique solution de l'équation précédente quelque soit k, est (0,0) mais, (0,0) n'appartient pas au domaine de f.

Le dernier tracé est presque un tracé parfait de la surface. Si le tracé avait été parfait, il aurait été moins évident de visualiser qu'il n'existe aucun voisinage de (0,0) pour lequel la fonction est toujours définie puisque les point manquants de cette surface le long de l'axe des z (et appartenant au plan $y=-x$) ne peuvent être visibles à l'écran. Un point étant de dimension nulle.

Évaluons cette limite et voyons le résultat que l'évaluateur donnera.

```
> limit(f(x,y),{x=0,y=0});
```

$$\textit{undefined} \quad \textbf{(4.2.1.10)}$$

Lorsque que l'évaluateur répond « *undefined* », c'est parce que la limite demandée dépend de la façon dont on s'approche de (0,0). Et donc, que la limite n'existe pas.

Cas où la limite n'existe pas par approche

Soit la fonction f définie par $f(x,y)=\frac{x^2}{x^2+y^2}$. Le domaine de la fonction f est $\mathbb{R} \times \mathbb{R} \setminus \{(0,0)\}$.
Dans ce cas, la fonction f est toujours définie dans n'importe quel voisinage de (0,0).

Demandons à l'évaluateur de calculer la limite de $f(x,y)$ lorsque $(x,y) \to (0,0)$.

```
> f:=(x,y)->x^2/(x^2+y^2);
```

$$f := (x,y) \mapsto \frac{x^2}{y^2+x^2} \quad \textbf{(4.2.2.1)}$$

```
> limit(f(x,y),{x=0,y=0});
```

$$0..1 \quad \textbf{(4.2.2.2)}$$

Lorsque que l'évaluateur répond « 0 ..1 », c'est parce que la limite demandée donnent des valeurs "oscillant" entre 0 et 1 selon la manière dont on s'approche de (0,0). Et donc, que la limite n'existe pas.

Traçons cette fonction pour $x \in [-0.25; 0.25]$ et pour $y \in [-0.25 ; 0.25]$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.25..0.25,y=-0.25..0.25,
              grid=[120,120],
              axes=normal,linestyle=2,orientation=[56,68,0],color=
  "Bright Cyan"):
> Vue_1:=display(Surface):
> display(Vue_1,shading=zhue);
```



Sur le graphique de la surface, on voit bien qu'avec une approche de (0,0) le long de l'axe des x ($y = 0$), les images $f(x, 0)$ par la fonction f tend vers 1 lorsque $x \rightarrow 0$.

En augmentant la résolution du tracé, cela devient davantage évident. (Essayez avec différentes valeurs de `grid`)

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.25..0.25,y=-0.25..0.25,
              grid=[240,240],axis=[tickmarks=4,color=gray],
              axes=normal,linestyle=2,orientation=[65,60,0],color=
  "Bright Cyan"):
> Vue_2:=display(Surface):
> display(Vue_2);
```



Illustrons les images de (x, y) avec le chemin $y = 0$, c'est-à-dire le long de l'axe des y.

```
> Vue_2:=display(Surface,transparency=0.3):
  Chemin:=spacecurve([t,0,0],t=0..0.25,
          color="Niagara Burgundy",
          thickness=5):
  Image:=spacecurve([t,0,f(t,0)],t=0..0.25,
          grid=[60,60],color="Niagara Burgundy",
          thickness=5):
```

```
> display([Vue_2,Chemin,Image],orientation=[65,65,0]);
```



Évaluons la limite avec l'approche $y = 0$.

```
> Limit(subs(y=0,f(x,y)),x=0)=limit(subs(y=0,f(x,y)),x=0);
```

$$\lim_{x \to 0} 1 = 1 \tag{4.2.2.3}$$

Alors, avec cette approche, $f(x, y) = \dfrac{x^2}{x^2 + y^2}$ se réduit à $\dfrac{x^2}{x^2}$, soit 1, et donc que la limite de $f(x, y)$ selon ce chemin est 1.

En modifiant l'orientation du graphique précédent, cela nous permet de constater qu'avec une approche le long de l'axe des y, $(x = 0)$, les images $f(0, y)$ par la fonction f tend vers 0 lorsque $y \rightarrow 0$.

```
> display(Vue_2,orientation=[30,65,0]);
```



Illustrons les images de (x,y) le long de l'axe des y avec le chemin $x = 0$.

```
> Chemin:=spacecurve([0,t,0],t=0..0.25,
           color="Niagara Burgundy",
           thickness=3):
  Image:=spacecurve([0,t,f(0,t)],t=0..0.25,
           color="Niagara Burgundy",
           thickness=3):
```

```
> display([Vue_2,Chemin,Image],orientation=[30,65,0]);
```



Évaluons la limite avec l'approche $x = 0$.

```
> Limit(subs(x=0,f(x,y)),y=0)=limit(subs(x=0,f(x,y)),y=0);
```

$$\lim_{y \to 0} 0 = 0 \quad \textbf{(4.2.2.4)}$$

Analytiquement, avec cette approche ($x = 0$), $f(x, y) = \frac{x^2}{x^2 + y^2}$ se réduit à 0 ($y \neq 0$), et donc que la limite de $f(x, y)$ selon ce chemin est 0.
On conclut donc que la limite demandée, la limite de $f(x, y)$ lorsque $(x, y) \to (0, 0)$, n'existe pas.

Il y a bien sûr d'autres chemins menant à l'origine; par exemple, (x, y) peut atteindre l'origine en suivant la droite $y = 2\,x$, ou bien en suivant la parabole $y = x^2$, ou bien en suivant la courbe $x = \arcsin(y)$, ou bien en suivant la parabole cubique $y = x^3$, ou bien en suivant le chemin $y = e^x - 1$, etc... Il y a une infinité de chemins passant par (0,0).

Illustrons les images de (x, y) avec le chemin $y = x$.

```
> Chemin:=spacecurve([t,t,0],t=0..0.25,
          color="Niagara Burgundy",
          thickness=3):
  Image:=spacecurve([t,t,f(t,t)],t=0..0.25,
          color="Niagara Burgundy",
          thickness=3):
```

```
> display([Vue_2,Chemin,Image],orientation=[15,70,0]);
```



Évaluons la limite avec l'approche $y = x$.

```
> Limit(subs(y=x,f(x,y)),x=0)=limit(subs(y=x,f(x,y)),x=0);
```

$$\lim_{x \to 0} \frac{1}{2} = \frac{1}{2} \tag{4.2.2.5}$$

Analytiquement, avec cette approche ($y = x$), $f(x, y) = \frac{x^2}{x^2 + y^2}$ se réduit à $\frac{x^2}{x^2 + x^2} = \frac{1}{2}$ et donc, que la limite de $f(x, y)$ selon ce chemin est $\frac{1}{2}$.

Allons-y avec un dernier chemin $y = e^x - 1$.

```
> Chemin:=spacecurve([t,exp(t)-1,0],t=0..0.2,
          color="Niagara Burgundy",
          thickness=3):
  Image:=spacecurve([t,exp(t)-1,f(t,exp(t)-1)],t=0..0.2,
          color="Niagara Burgundy",
          thickness=3):
```

```
> display([Vue_2,Chemin,Image],orientation=[15,65,0]);
```



Évaluons la limite avec l'approche $y = e^x - 1$.

```
> Limit(subs(y=exp(x)-1,f(x,y)),x=0)=limit(subs(y=exp(x)-1,f(x,y)),x=0);
```

$$\lim_{x \to 0} \frac{x^2}{x^2 + (e^x - 1)^2} = \frac{1}{2} \qquad \textbf{(4.2.2.6)}$$

Analytiquement, avec cette approche ($y = e^x - 1$), $f(x, y) = \dfrac{x^2}{x^2 + y^2}$ se réduit à $\dfrac{x^2}{x^2 + (e^x - 1)^2}$ et il est moins évident algébriquement (indétermination du type $\dfrac{0}{0}$) que la limite de $f(x, y)$ selon ce chemin est $\dfrac{1}{2}$.

Dérivation d'une fonction d'une seule variable

Il y a deux manières d'opérer la dérivation. Selon que l'on dérive une fonction ou selon que l'on dérive une expression. Pour obtenir la **fonction dérivée** par la dérivation d'une fonction, il faut utiliser l'opérateur de dérivation **D**.

Opérateur de dérivation D

```
> f:=x->(x^4+1)^7;
```

(5.1.1)

$$f := x \mapsto \left(x^4+1\right)^7 \quad \textbf{(5.1.1)}$$

```
> D(f)(x);
```

$$28\left(x^4+1\right)^6 x^3 \quad \textbf{(5.1.2)}$$

La documentation de la feuille de travail gagne en clarté en utilisant les apostrophes droits:

```
> 'D'(f)(x)=D(f)(x);
  'D'(f)(sqrt(2))=D(f)(sqrt(2));
```

$$\mathrm{D}(f)(x) = 28\left(x^4+1\right)^6 x^3$$

$$\mathrm{D}(f)\left(\sqrt{2}\right) = 875000\sqrt{2} \quad \textbf{(5.1.3)}$$

Pour obtenir des fonctions dérivées successives d'ordre supérieur, il faut utiliser l'opérateur de composition réitérée @@.

```
> '(D@@4)'(f)(x)=(D@@4)(f)(x);
```

$$\mathrm{D}^{(4)}(f)(x) = 215040\left(x^4+1\right)^3 x^{12} + 241920\left(x^4+1\right)^4 x^8 + 34272\left(x^4+1\right)^5 x^4 + 168\left(x^4+1\right)^6 \quad \textbf{(5.1.4)}$$

```
> ``=expand(rhs((5.1.4)));
```

$$= 491400x^{24} + 1785168x^{20} + 2441880x^{16} + 1528800x^{12} + 415800x^8 + 35280x^4 + 168 \quad \textbf{(5.1.5)}$$

```
> f:='f':
```

L'opérateur de dérivation **D** est utilisé pour obtenir la fonction dérivée tandis que la macro-commande **diff** est utilisée pour dériver une expression. La macro-commande **diff** s'applique donc à une formule et a comme résultat une formule (et non pas une fonction).

Macro-commande diff

```
> y:=(x^3-4*x)*(x^4-5*x^2+3*x);
```

$$y := \left(x^3-4x\right)\left(x^4-5x^2+3x\right) \quad \textbf{(5.2.1)}$$

```
> diff(y,x);
```

$$\left(3x^2-4\right)\left(x^4-5x^2+3x\right) + \left(x^3-4x\right)\left(4x^3-10x+3\right) \quad \textbf{(5.2.2)}$$

```
> factor((5.2.2));
```

$$x\left(7x^5-45x^3+12x^2+60x-24\right) \quad \textbf{(5.2.3)}$$

```
> eval((5.2.3),x=-2);
```

$$-80 \quad \textbf{(5.2.4)}$$

La macro-commande **diff** possède une forme inerte qui est bien utile pour une bonne documentation du développement des calculs.

```
> y:=(cos^2)(theta)*sin(theta);
```

$$y := \cos(\theta)^2 \sin(\theta) \quad \textbf{(5.2.5)}$$

```
> Diff(y,theta)=diff(y,theta);
```

$$\frac{d}{d\theta}\left(\cos(\theta)^2\sin(\theta)\right) = -2\cos(\theta)\sin(\theta)^2 + \cos(\theta)^3 \quad \textbf{(5.2.6)}$$

```
> ``=simplify(rhs((5.2.6)),trig);
```

$$= 3\cos(\theta)^3 - 2\cos(\theta) \quad \textbf{(5.2.7)}$$

```
> Eval(rhs((5.2.7)),theta=Pi/4)=eval(rhs((5.2.7)),theta=Pi/4);
```

$$\left(3\cos(\theta)^3 - 2\cos(\theta)\right)\Big|_{\theta=\frac{\pi}{4}} = -\frac{\sqrt{2}}{4} \tag{5.2.8}$$

On peut aussi utiliser les accents graves (backquotes) pour la création de noms afin de bien documenter un développement.

```
> y:=(3*x^3-1)^6;
```

$$y := \left(3x^3 - 1\right)^6 \tag{5.2.9}$$

```
> `y'`:=diff(y,x);
```

$$y' := 54\left(3x^3 - 1\right)^5 x^2 \tag{5.2.10}$$

```
> Eval('`y'`',x=-sqrt(2))=eval(`y'`,x=-sqrt(2));
```

$$y'\Big|_{x=-\sqrt{2}} = 108\left(-6\sqrt{2} - 1\right)^5 \tag{5.2.11}$$

```
> ``=radnormal(rhs((5.2.11)));
```

$$= -2877228 - 3829032\sqrt{2} \tag{5.2.12}$$

Pour obtenir une dérivée successive d'ordre *n*, il suffit, dans la macro-commande diff , de répéter la variable autant de fois qu'il faut selon l'ordre de dérivation voulu.

```
> y:=1/sqrt(x);
```

$$y := \frac{1}{\sqrt{x}} \tag{5.2.13}$$

```
> `y'''`:=diff(y,x,x,x);
```

$$y''' := -\frac{15}{8x^{7/2}} \tag{5.2.14}$$

Pour préciser de manière plus concise l'ordre de la dérivée successive, on peut utiliser l'opérateur de séquence $.

```
> y:=x/(x+1);
```

$$y := \frac{x}{x+1} \tag{5.2.15}$$

```
> Diff(y,x$6)=diff(y,x$6);
```

$$\frac{d^6}{dx^6}\left(\frac{x}{x+1}\right) = -\frac{720}{(x+1)^6} + \frac{720x}{(x+1)^7} \tag{5.2.16}$$

```
> ``=normal(rhs((5.2.16)));
```

$$= -\frac{720}{(x+1)^7} \tag{5.2.17}$$

Dérivation d'équations implicites d'une fonction d'une seule variable

Il y a deux manières de dériver implicitement:

- avec la macro-commande diff. Dans ce cas, il faut, dans l'équation, **préciser la variable dépendante en utilisant, pour celle-ci, la syntaxe fonctionnelle**.
- avec la macro-commande implicitdiff.

```
> x:='x': y:='y': `y'`:='`y'`':
Équation:=3*y(x)^2+5*x=3-5*y(x)^3;
```

$$Équation := 3\,y(x)^2 + 5\,x = 3 - 5\,y(x)^3 \quad (6.1)$$

```
> Équation_dérivée:=diff(Équation,x);
```

$$Équation\_dérivée := 6\,y(x)\left(\frac{d}{dx}\,y(x)\right) + 5 = -15\,y(x)^2\left(\frac{d}{dx}\,y(x)\right) \quad (6.2)$$

Isolons maintenant la variable dépendante.

```
> solve(Équation_dérivée,{diff(y(x),x)});
```

$$\left\{\frac{d}{dx}\,y(x) = -\frac{5}{3\,y(x)\,(5\,y(x)+2)}\right\} \quad (6.3)$$

```
> isolate(Équation_dérivée,diff(y(x),x));
```

$$\frac{d}{dx}\,y(x) = -\frac{5}{6\,y(x) + 15\,y(x)^2} \quad (6.4)$$

```
> ``=factor(rhs((6.4)));
```

$$= -\frac{5}{3\,y(x)\,(5\,y(x)+2)} \quad (6.5)$$

Dérivons maintenant avec la macro-commande **implicitdiff**. Dans ce cas, *ne pas utiliser la syntaxe fonctionnelle pour la variable dépendante.*

```
> Équation:=3*y^2+5*x=3-5*y^3;
```

$$Équation := 3\,y^2 + 5\,x = -5\,y^3 + 3 \quad (6.6)$$

```
> implicitdiff(Équation,y,x);
```

$$-\frac{5}{3\,y\,(5\,y+2)} \quad (6.7)$$

```
> `y'`=implicitdiff(Équation,y,x);
```

$$y' = -\frac{5}{3\,y\,(5\,y+2)} \quad (6.8)$$

```
> `y'`:='`y'`':
```

L'intégration d'une fonction d'une seule variable

```
> Int(f(x),x);
```

$$\int f(x)\,dx \quad (7.1)$$

On ne présente ici que la syntaxe de l'intégrale indéfinie et définie sans explorer plus avant certaines techiques d'intégration.

Calculons $\int \frac{x-3}{x^2+4}\,dx$.

```
> f:=x->(x-3)/(x^2+4);
```

(7.2)

$$f := x \mapsto \frac{x-3}{x^2+4} \tag{7.2}$$

```
> Int(f(x),x)=int(f(x),x);
```

$$\int \frac{x-3}{x^2+4}\,dx = \frac{\ln(x^2+4)}{2} - \frac{3\arctan\left(\frac{x}{2}\right)}{2} \tag{7.3}$$

Notons que Maple pose automatiquement la constante d'intégration égale à 0. Pour contourner ce choix de programmation, incorporons au départ la constante C d'intégration.

La forme inerte **Int** est bien utile à une bonne documentation dans un développement.

```
> Int(f(x),x)=int(f(x),x)+C;
```

$$\int \frac{x-3}{x^2+4}\,dx = \frac{\ln(x^2+4)}{2} - \frac{3\arctan\left(\frac{x}{2}\right)}{2} + C \tag{7.4}$$

Il peut arriver que Maple emploi des fonctions mathématiques inconnues du niveau collégial dans la formulation de l'antidérivée.

```
> f:=x->sqrt(1-x^4);
```

$$f := x \mapsto \sqrt{1-x^4} \tag{7.5}$$

```
> Int(f(x),x)=int(f(x),x)+C;
```

$$\int \sqrt{-x^4+1}\,dx = \frac{x\sqrt{-x^4+1}}{3} + \frac{2\sqrt{-x^2+1}\,\sqrt{x^2+1}\,\mathrm{EllipticF}(x, \mathrm{I})}{3\sqrt{-x^4+1}} + C \tag{7.6}$$

Pas de panique, un choix judicieux des intégrales indéfinies sera fait dans votre cours pour éviter de telles (bonnes) réponses.

Il est même possible que Maple ne puisse automatiquement faire l'intégration indéfinie. Dans ce cas, l'évaluateur retourne, comme un écho, l'intégrale demandée.

```
> f:=x->ln(sin(x))/x;
```

$$f := x \mapsto \frac{\ln(\sin(x))}{x} \tag{7.7}$$

```
> Int(f(x),x)=int(f(x),x);
```

$$\int \frac{\ln(\sin(x))}{x}\,dx = \int \frac{\ln(\sin(x))}{x}\,dx \tag{7.8}$$

Remarquez, dans la réponse, la différence de notation du membre de gauche et du membre de droite de l'égalité. **Dans le membre de gauche, les morceaux en gris pâles indiquent que c'est l'usager qui a utilisé la forme inerte de la macro-commande** tandis que dans le membre de droite, tout en bleue, indique que l'évaluateur n'a pu résoudre la requête.

Autres exemples:

Évaluons l'intégrale définie: $\displaystyle\int_0^{\frac{1}{2}} \frac{1}{\sqrt{1-x^2}}\,dx$

```
> Int(1/sqrt(1-x^2),x=0..1/2)=int(1/sqrt(1-x^2),x=0..1/2);
```

(7.9)

$$\int_0^{\frac{1}{2}} \frac{1}{\sqrt{-x^2+1}}\,dx = \frac{\pi}{6} \tag{7.9}$$

Soit le calcul de $\int_0^{\pi} (2+\cos(\theta))^2\, d\theta$

```
> f:=theta->(2+cos(theta))^2;
```

$$f := \theta \mapsto (2+\cos(\theta))^2 \tag{7.10}$$

```
> Int(f(theta),theta=0..Pi)=int(f(theta),theta=0..Pi);
```

$$\int_0^{\pi} (2+\cos(\theta))^2\, d\theta = \frac{9\pi}{2} \tag{7.11}$$

Soit le calcul de l'intégrale impropre $\int_1^{\infty} \frac{1}{x^2}\, dx$

```
> f:=x->1/x^2;
```

$$f := x \mapsto \frac{1}{x^2} \tag{7.12}$$

```
> Int(f(x),x=1..infinity)=int(f(x),x=1..infinity);
```

$$\int_1^{\infty} \frac{1}{x^2}\, dx = 1 \tag{7.13}$$

Dérivées partielles d'une fonction explicite de deux variables

Avec l'opérateur de dérivation D

Soit la fonction f définie par $f(x,y) = \ln(\sin(x^2) + 2y^2)$.

```
> f:=(x,y)-> ln(sin(x^2)+2*y^2);
```

$$f := (x,y) \mapsto \ln(\sin(x^2) + 2 \cdot y^2) \tag{8.1.1}$$

La fonction dérivée partielle par rapport à x, $\frac{\partial}{\partial x} f$, est obtenue ainsi:

```
> D[1](f);
```

$$(x,y) \mapsto \frac{2 \cdot x \cdot \cos(x^2)}{\sin(x^2) + 2 \cdot y^2} \tag{8.1.2}$$

Alors, l'évaluation au point $\left(\frac{\pi}{2}, \frac{\pi}{4}\right)$ de la fonction dérivée partielle par rapport à x s'écrit

```
> 'D[1](f)'(Pi/2,Pi/4);
```

$$D_1(f)\left(\frac{\pi}{2}, \frac{\pi}{4}\right) \tag{8.1.3}$$

```
> 'D[1](f)'(Pi/2,Pi/4)=D[1](f)(Pi/2,Pi/4);
```

$$D_1(f)\left(\frac{\pi}{2}, \frac{\pi}{4}\right) = \frac{\pi \cos\left(\frac{\pi^2}{4}\right)}{\sin\left(\frac{\pi^2}{4}\right) + \frac{\pi^2}{8}} \tag{8.1.4}$$

Tandis que la **formule** de la dérivée partielle par rapport à x de la fonction f:

```
> D[1](f)(x,y);
```

$$\frac{2 x \cos(x^2)}{\sin(x^2) + 2 y^2} \tag{8.1.5}$$

L'évaluation au point $\left(\frac{\pi}{2}, \frac{\pi}{4}\right)$ de la fonction dérivée partielle par rapport à x se pose alors plus laborieusement

```
> Eval(D[1](f)(x,y),[x=Pi/2,y=Pi/4]);
```

$$\left.\frac{2 x \cos(x^2)}{\sin(x^2) + 2 y^2}\right|_{\left[x=\frac{\pi}{2}, y=\frac{\pi}{4}\right]} \tag{8.1.6}$$

```
> Eval(D[1](f)(x,y),[x=Pi/2,y=Pi/4])=eval(D[1](f)(x,y),[x=Pi/2,y=
  Pi/4]);
```

$$\left.\frac{2 x \cos(x^2)}{\sin(x^2) + 2 y^2}\right|_{\left[x=\frac{\pi}{2}, y=\frac{\pi}{4}\right]} = \frac{\pi \cos\left(\frac{\pi^2}{4}\right)}{\sin\left(\frac{\pi^2}{4}\right) + \frac{\pi^2}{8}} \tag{8.1.7}$$

On a précisé, à l'opérateur **D**, le nombre 1 entre crochet car, au moment de la création de la fonction f, la première variable précisée a été la variable x. Alors, la formule de la dérivée partielle par rapport à y, $\frac{\partial}{\partial y} f$, sera

```
> D[2](f)(x,y);
```

$$\frac{4 y}{\sin(x^2) + 2 y^2} \tag{8.1.8}$$

Pour le calcul des dérivées mixtes, il suffit de préciser entre crochets l'ordre des variables de la dérivation.

Par exemple, trouver $\frac{\partial^2}{\partial y \, \partial x} f$

```
> D[2,1](f)(x,y);
```

$$-\frac{8 y x \cos(x^2)}{\left(\sin(x^2) + 2 y^2\right)^2} \tag{8.1.9}$$

En fait, **D[i,j](f)** correspond à **D[i](D[j](f))**.

Alors, le calcul de $\frac{\partial^2}{\partial x \, \partial y} f(x, y)$ est posé comme suit:

```
> D[1,2](f)(x,y);
```

(8.1.10)

$$-\frac{8\,y\,x\cos(x^2)}{(\sin(x^2)+2y^2)^2} \qquad \textbf{(8.1.10)}$$

Pourquoi a-t-on $\frac{\partial^2}{\partial y\,\partial x} f(x,y) = \frac{\partial^2}{\partial x\,\partial y} f(x,y)$? En est-il toujours ainsi ?

Pour calculer la dérivée partielle (par rapport à x) successive d'ordre 3, $\frac{\partial^3}{\partial x^3} f(x,y)$, on devra répéter trois fois le chiffre 1:

```
> D[1,1,1](f)(x,y);
```

$$-\frac{12\,x\sin(x^2)}{\sin(x^2)+2y^2} - \frac{12\cos(x^2)^2\,x}{(\sin(x^2)+2y^2)^2} - \frac{8x^3\cos(x^2)}{\sin(x^2)+2y^2} + \frac{24\,x^3\sin(x^2)\cos(x^2)}{(\sin(x^2)+2y^2)^2} + \frac{16\,x^3\cos(x^2)^3}{(\sin(x^2)+2y^2)^3} \qquad \textbf{(8.1.11)}$$

On peut, bien sûr, employer l'opérateur de création de séquence **$**.

```
> D[$(1,3)](f)(x,y);
```

$$-\frac{12\,x\sin(x^2)}{\sin(x^2)+2y^2} - \frac{12\cos(x^2)^2\,x}{(\sin(x^2)+2y^2)^2} - \frac{8x^3\cos(x^2)}{\sin(x^2)+2y^2} + \frac{24\,x^3\sin(x^2)\cos(x^2)}{(\sin(x^2)+2y^2)^2} + \frac{16\,x^3\cos(x^2)^3}{(\sin(x^2)+2y^2)^3} \qquad \textbf{(8.1.12)}$$

Voici un autre exemple de calcul.
Trouver la dérivée partielle d'ordre 5 $\frac{\partial^5}{\partial y^3\,\partial x^2} f(x,y)$.

```
> 'D[$(2,3),$(1,2)](f)(x,y)';
```

$$D_{\underbrace{2,\,\ldots,\,2}_{3\text{ times}},\,\underbrace{1,\,\ldots,\,1}_{2\text{ times}}}(f)(x,y) \qquad \textbf{(8.1.13)}$$

Avec la macro-commande de dérivation diff

Soit à calculer $\frac{\partial}{\partial x}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right)$.

Afin de documenter plus clairement les résultats, utilisons la forme inactive de diff, soit `Diff` et formulons les requêtes sous la forme d'une équation

Forme inerte = Forme active

```
> Diff(sqrt(x^2-y^2)/(x-y),x)=diff(sqrt(x^2-y^2)/(x-y),x);
```

$$\frac{\partial}{\partial x}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right) = \frac{x}{\sqrt{x^2-y^2}\,(x-y)} - \frac{\sqrt{x^2-y^2}}{(x-y)^2} \qquad \textbf{(8.2.1)}$$

Simplifions ce résultat en normalisant la soustraction des deux fractions.

```
> normal((8.2.1));
```

$$\frac{\partial}{\partial x}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right) = -\frac{y}{\sqrt{x^2-y^2}\,(x-y)} \qquad \textbf{(8.2.2)}$$

Soit à calculer $\frac{\partial^2}{\partial y\,\partial x}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right)$.

```
> Diff(sqrt(x^2-y^2)/(x-y),x,y)=diff(sqrt(x^2-y^2)/(x-y),x,y);
```

$$\frac{\partial^2}{\partial x\partial y}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right) = \frac{xy}{(x^2-y^2)^{3/2}(x-y)} + \frac{x}{\sqrt{x^2-y^2}\,(x-y)^2} + \frac{y}{\sqrt{x^2-y^2}\,(x-y)^2} - \frac{2\sqrt{x^2-y^2}}{(x-y)^3} \qquad \textbf{(8.2.3)}$$

Simplifions.

```
> normal((8.2.3));
```

$$\frac{\partial^2}{\partial x\partial y}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right) = -\frac{x^2+xy+y^2}{(x^2-y^2)^{3/2}(x-y)} \qquad \textbf{(8.2.4)}$$

En fait, **diff(Formule, x, y) est équivalent à diff(diff (Formule, x), y)**.

Soit à calculer $\frac{\partial^3}{\partial y^2\,\partial x}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right)$

```
> Diff(sqrt(x^2-y^2)/(x-y),$(y,2),x)=diff(sqrt(x^2-y^2)/(x-y),$(y,
  2),x);
```

$$\frac{\partial^3}{\partial y^2\partial x}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right) = \frac{x}{(x^2-y^2)^{3/2}(x-y)} + \frac{1}{\sqrt{x^2-y^2}\,(x-y)^2} + \frac{3y^2x}{(x^2-y^2)^{5/2}(x-y)} + \frac{y^2}{(x^2-y^2)^{3/2}(x-y)^2} + \frac{2yx}{(x^2-y^2)^{3/2}(x-y)^2} + \frac{4y}{\sqrt{x^2-y^2}\,(x-y)^3} + \frac{2x}{\sqrt{x^2-y^2}\,(x-y)^3} - \frac{6\sqrt{x^2-y^2}}{(x-y)^4} \qquad \textbf{(8.2.5)}$$

Simplifions.

```
> normal((8.2.5));
```

$$\frac{\partial^3}{\partial y^2\partial x}\left(\frac{\sqrt{x^2-y^2}}{x-y}\right) = -\frac{2x^3+7yx^2+4y^2x+2y^3}{(x^2-y^2)^{5/2}(x-y)} \qquad \textbf{(8.2.6)}$$

Dérivées partielles d'une fonction implicite de deux variables

Soit $z = f(x, y)$ où z est définie implicitement par la relation $xy^2 + xyz = 2 - z^3$. Pour trouver $\frac{\partial}{\partial x} z$, il faut dériver chaque membre de cette équation, par rapport à x. Dans ce cas, il faut signifier à l'évaluateur que z est définie implicitement en termes de x et y. Pour signifier à l'évaluateur que z est une variable dépendante de x et de y, il faut utiliser la syntaxe fonctionnelle $z(x, y)$ au lieu de saisir seulement z.

Créons l'équation à dériver.

```
> Éq:= x*y^2+x*y*z(x,y) = 2-z(x,y)^3;
```

$$Éq := y^2 x + x y z(x,y) = 2 - z(x,y)^3 \tag{9.1}$$

Dérivons ensuite chaque membre de cette équation par rapport à x.

```
> Éq_dérivée:=diff(Éq,x);
```

$$Éq\_dérivée := y^2 + y z(x,y) + x y \left(\frac{\partial}{\partial x} z(x,y)\right) = -3 z(x,y)^2 \left(\frac{\partial}{\partial x} z(x,y)\right) \tag{9.2}$$

Reste donc à isoler $\frac{\partial}{\partial x} z(x,y)$.

Employons la macro-commande **`isolate`**.

```
> isolate(Éq_dérivée,diff(z(x,y),x));
```

$$\frac{\partial}{\partial x} z(x,y) = \frac{-y^2 - y z(x,y)}{x y + 3 z(x,y)^2} \tag{9.3}$$

Simplifions en factorisant ce résultat.

```
> ``=factor(rhs((9.3)));
```

$$= -\frac{y\,(z(x,y) + y)}{x y + 3 z(x,y)^2} \tag{9.4}$$

Pour obtenir plus directement la formule de $\frac{\partial}{\partial x} z$, il est de loin préférable **d'employer la macro-commande** `implicitdiff`.

Reformulons **Éq** en terme de z et non pas en terme de $z(x,y)$.

```
> Éq:= x*y^2+x*y*z = 2-z^3;
```

$$Éq := y^2 x + x y z = -z^3 + 2 \tag{9.5}$$

Le second argument de la macro-commande **`implicitdiff`** doit obligatoirement préciser lesquelles des variables en causes sont les variables dépendantes. On emploiera pour cela la syntaxe fonctionnelle.

```
> Diff(z,x)=implicitdiff(Éq,z(x,y),x);
```

$$\frac{\partial}{\partial x} z = -\frac{y\,(z + y)}{x y + 3 z^2} \tag{9.6}$$

Le résultat précédent est plus conforme à la notation habituelle utilisée en classe.

Comme dernier exemple, trouvons $\frac{\partial^2}{\partial x^2} z$.

```
> Diff(z,$(x,2))=implicitdiff(Éq,z(x,y),$(x,2));
```

$$\frac{\partial^2}{\partial x^2} z = -\frac{2 y^3 \left(-x y - z x + 3 y z + 3 z^2\right)}{x^3 y^3 + 9 z^2 x^2 y^2 + 27 z^4 x y + 27 z^6} \tag{9.7}$$

Essayons de simplifier ce résulat en factorisant.

```
> ``=factor(rhs((9.7)));
```

$$= \frac{2 y^3 (z + y)(x - 3 z)}{(x y + 3 z^2)^3} \tag{9.8}$$

L'intégration d'une fonction de deux variables

Le calcul d'une intégrale double est, comme vous le savez, un calcul successif de deux intégrales simples.

Soit le calcul de $\int_3^4 \int_1^2 \frac{1}{(x+y)^2}\, dy\, dx$.

Une manière d'effectuer le calcul demandé est de calculer successivement les deux intégrales simples en commençant par l'intégrale imbriquée. Il faut donc intégrer d'abord par rapport à la variable *y*.
Posons le calcul demandé.

```
> Calcul:=Int(Int(1/((x+y)^2),y = 1 .. 2),x = 3 .. 4);
```

$$Calcul := \int_3^4 \int_1^2 \frac{1}{(x+y)^2}\, dy\, dx \qquad \textbf{(10.1)}$$

Intégrons d'abord par rapport à *y*. Posons ce calcul.

```
> yintégrale:=Int(1/(x+y)^2,y=1..2);
```

$$yintégrale := \int_1^2 \frac{1}{(x+y)^2}\, dy \qquad \textbf{(10.2)}$$

Imbriquons ensuite l'évaluation de *yintegrale* dans l'intégration par rapport à *x*.

```
> Int(value(yintégrale),x=3..4);
```

$$\int_3^4 \left(\left(\begin{cases} \infty & -2 < x \wedge x < -1 \\ 0 & otherwise \end{cases} \right) + \frac{1}{(x+1)\,(x+2)} \right) dx \qquad \textbf{(10.3)}$$

Reste donc à évaluer la seconde intégrale par rapport à *x*.

```
> Calcul=value((10.3));
```

$$\int_3^4 \int_1^2 \frac{1}{(x+y)^2}\, dy\, dx = -3\ln(2) + 2\ln(5) - \ln(3) \qquad \textbf{(10.4)}$$

Une autre façon de faire est d'imbriquer immédiatement ces deux intégrales simples. Cette manière s'avère plus directe et donc plus claire.

```
> Calcul=int(int(1/(x+y)^2,y=1..2),x=3..4);
```

$$\int_3^4 \int_1^2 \frac{1}{(x+y)^2}\, dy\, dx = -3\ln(2) + 2\ln(5) - \ln(3) \qquad \textbf{(10.5)}$$

Simplifions l'écriture logarithmique de ce résultat à l'aide des propriétés des logarithmes.

```
> ``=combine(rhs((10.5)),ln);
```

$$= -6\ln\left(\frac{\sqrt{2}\, 5^{2/3}\, 3^{1/6}}{5} \right) \qquad \textbf{(10.6)}$$

Dans une intégrale double, les bornes d'intégration de l'intégrale imbriquée ne sont pas nécessairement des constantes.

Calculer $\int_0^{\pi} \int_0^{3\sin(\theta)} r\sqrt{9-r^2}\, dr\, d\theta$.

Posons le calcul demandé.

```
> Calcul:=Int(Int(r*sqrt(9-r^2),r=0..3*sin(theta)),theta=0..Pi):
  Calcul;
```

$$\int_0^{\pi}\int_0^{3\sin(\theta)} r\sqrt{-r^2+9}\,\mathrm{d}r\,\mathrm{d}\theta \tag{10.7}$$

Évaluons.

```
> Calcul=int(int(r*sqrt(9-r^2),r=0..3*sin(theta)),theta=0..Pi);
```

$$\int_0^{\pi}\int_0^{3\sin(\theta)} r\sqrt{-r^2+9}\,\mathrm{d}r\,\mathrm{d}\theta = -12+9\pi \tag{10.8}$$

Voici un autre exemple de calcul d'une intégrale double. Calculer $\int_2^{2\sqrt{2}}\int_{\arccos\left(\frac{2}{r}\right)}^{\arcsin\left(\frac{2}{r}\right)} r\,\mathrm{d}\theta\,\mathrm{d}r$.

Posons le calcul demandé.

```
> Calcul:=Int(Int(r,theta=arccos(2/r)..arcsin(2/r)),r=2..2*sqrt(2)):
  Calcul;
```

$$\int_2^{2\sqrt{2}}\int_{\arccos\left(\frac{2}{r}\right)}^{\arcsin\left(\frac{2}{r}\right)} r\,\mathrm{d}\theta\,\mathrm{d}r \tag{10.9}$$

Évaluons.

```
> Calcul=value(Calcul);
```

$$\int_2^{2\sqrt{2}}\int_{\arccos\left(\frac{2}{r}\right)}^{\arcsin\left(\frac{2}{r}\right)} r\,\mathrm{d}\theta\,\mathrm{d}r = 4-\pi \tag{10.10}$$

Les exemples précédents ont montré comment employer Maple dans des calculs formels. On peut, bien sûr, obtenir une approximation numérique plutôt qu'une évaluation symbolique (exacte) lorsqu'on évalue des intégrales définies. Dans ce cas, il faut employer la macro-commande `evalf`.

Obtenez une approximation de $\int_{\frac{\pi}{16}}^{\frac{\pi}{8}}\int_1^{\frac{16\,\theta}{\pi}} r\,\mathrm{d}r\,\mathrm{d}\theta$.

Posons le calcul demandé.

```
> Calcul:=Int(Int(r,r=1..16*theta/Pi),theta=Pi/16..Pi/8);
```

$$Calcul := \int_{\frac{\pi}{16}}^{\frac{\pi}{8}}\int_1^{\frac{16\,\theta}{\pi}} r\,\mathrm{d}r\,\mathrm{d}\theta \tag{10.11}$$

Obtenons une approximation avec des calculs impliquant 20 décimales.

```
> Calcul=evalf[20](Calcul);
```

$$\int_{\frac{\pi}{16}}^{\frac{\pi}{8}} \int_{1}^{\frac{16\,\theta}{\pi}} r \,\mathrm{d}r\,\mathrm{d}\theta = 0.13089969389957471825 \qquad \textbf{(10.12)}$$

Comme dernier exemple, reprenons le calcul de l'exemple 1 de la page 89 du cahier de notes « Intégrales doubles » . Dans cet exemple, il est demandé d'intégrer $\int_0^2 \int_0^x \sqrt{x^2+y^2}\,\mathrm{d}y\,\mathrm{d}x$ par un changement de variables en coordonnées polaires selon les deux ordres $\mathrm{d}\theta\,\mathrm{d}r$ et $\mathrm{d}r\,\mathrm{d}\theta$. Selon l'ordre $\mathrm{d}\theta\,\mathrm{d}r$, on obtient $\int_0^2 \int_0^{\frac{\pi}{4}} r^2\,\mathrm{d}\theta\,\mathrm{d}r + \int_2^{2\sqrt{2}} \int_{\arccos\left(\frac{2}{r}\right)}^{\frac{\pi}{4}} r^2\,\mathrm{d}\theta\,\mathrm{d}r$ et le calcul est développé sur environ deux pages et demie, et où on applique les techniques d'intégration par parties (2 fois) et la substitution trigonométrique.

Voyons le résultat que donnera Maple. Posons le calcul demandé.

```
> Calcul:=Int(Int(r^2,theta = 0 .. Pi/4),r = 0 .. 2)+Int(Int(r^2,theta
  = arccos(2/r) .. Pi/4),r = 2 .. 2*sqrt(2));
```

$$Calcul := \int_0^2 \int_0^{\frac{\pi}{4}} r^2\,\mathrm{d}\theta\,\mathrm{d}r + \int_2^{2\sqrt{2}} \int_{\arccos\left(\frac{2}{r}\right)}^{\frac{\pi}{4}} r^2\,\mathrm{d}\theta\,\mathrm{d}r \qquad \textbf{(10.13)}$$

Évaluons.

```
> Calcul=value(Calcul);
```

$$\int_0^2 \int_0^{\frac{\pi}{4}} r^2\,\mathrm{d}\theta\,\mathrm{d}r + \int_2^{2\sqrt{2}} \int_{\arccos\left(\frac{2}{r}\right)}^{\frac{\pi}{4}} r^2\,\mathrm{d}\theta\,\mathrm{d}r = \frac{4\sqrt{2}}{3} + \frac{4\,\mathrm{arccoth}\left(\sqrt{2}\right)}{3} \qquad \textbf{(10.14)}$$

La rapidité du résultat est remarquable.

Le résultat précédent $\frac{4\sqrt{2}}{3} + \frac{4\,\mathrm{arccoth}\left(\sqrt{2}\right)}{3}$ qu'a donné l'évaluateur fait intervenir la fonction arc cotangente hyperbolique. Ce n'est pas une fonction usuelle pour le niveau du cours. Tout de même, montrons que cette notation symbolique est égale à celle obtenue dans notre cahier de notes $\frac{4\left(\sqrt{2}+\ln\left(\sqrt{2}+1\right)\right)}{3}$.

```
> Réponse_cahier:=4*(sqrt(2)+ln(sqrt(2)+1))/3;
```

$$Réponse\_cahier := \frac{4\sqrt{2}}{3} + \frac{4\ln\left(1+\sqrt{2}\right)}{3} \qquad \textbf{(10.15)}$$

Nous ne passerons par la définition de la fonction arccoth mais par une comparaison *numérique* de $\frac{4\sqrt{2}}{3} + \frac{4\,\mathrm{arccoth}\left(\sqrt{2}\right)}{3}$ avec la réponse du cahier $\frac{4\left(\sqrt{2}+\ln\left(\sqrt{2}+1\right)\right)}{3}$

```
> evalf[20](Calcul);
```

$$3.0607828658568507654 \qquad \textbf{(10.16)}$$

```
> evalf[20](Réponse_cahier);
```

$$3.0607828658568507653 \quad (10.17)$$

Il y a tout lieu de croire que la fonction ln est impliquée dans la définition de arccoth.

```
> convert(rhs((10.14)),ln);
```

$$\frac{4\sqrt{2}}{3} + \frac{2\ln\left(1+\sqrt{2}\right)}{3} - \frac{2\ln\left(\sqrt{2}-1\right)}{3} \quad (10.18)$$

```
> simplify((10.18));
```

$$\frac{4\sqrt{2}}{3} + \frac{4\ln\left(1+\sqrt{2}\right)}{3} \quad (10.19)$$